

Paper 165-2013

Clinician Prescribing Feedback Site: Comparing Clinician Prescribing Habits and Providing Actionable Patient Lists

Michael K. Nash, Kaiser Permanente Northwest

ABSTRACT

Which doctor is prescribing the most non-formulary medications? Which patients are on a brand drug when an equivalent generic is available? These questions and many more can be answered when using the Clinician Prescribing Feedback Site. This secure intranet site at Kaiser Permanente Northwest uses SAS®/GRAPH HBAR and VBAR charts to compare clinician prescribing habits. Drill down to compare all clinics, or all departments, or all doctors within a Clinic/Specialty. Drill down even further to find patient lists so pharmacists or clinician staff can perform outreach to members. The Clinician Prescribing Feedback Site also tracks costs and patients month to month. This paper will show you how to create linked html files using SAS PROC GCHART and the HTML= option.

INTRODUCTION

The Kaiser Permanente Northwest region (KPNW) is a non-profit integrated health care delivery system centered in Portland, Oregon with a service area ranging from Salem, OR in the south to Longview, WA in the north. KPNW has nearly 500,000 members in that service area with about 1,500 medical providers, and over two dozen primary and specialty care medical offices to serve our members.

The cost of pharmaceutical medication is a driving force in the increasing expense for health care in the U.S. today. In order to mitigate those costs, the pharmacy department of KPNW has been tasked with finding savings where possible. The greatest bulk of savings comes when a brand product switches to generic. These costs savings can be significant for the region and KPNW quickly works to migrate those members on a brand drug to the generic equivalent. Another form of savings comes from identifying members who are eligible to switch from a non-preferred or non-formulary medication to a preferred or formulary medication. The Clinician Prescribing Feedback Site is one of the tools that KPNW can employ to find those types of members.

The Clinician Prescribing Feedback Site is a secure intranet web site where approved users (doctors, pharmacists, and managers) can login and compare prescribing habits across the region, across a medical office clinic, or across a specialty. The power of the tool is in the ability for pharmacists, who work with a certain doctor, to drill down to find members who may warrant a medication change. The patient's doctor has the final say as to whether a medication change is warranted or not; the doctor then has a conversation with the patient about any possible medication changes.

SECURE SITE

Users of Clinician Prescribing Feedback Site must login into a secure KPNW intranet site. Access is restricted to clinicians, pharmacists, and pharmacy managers in order to protect Personal Health Information (PHI). To login, users go to the intranet site (Figure 1) and enter their username and password. Once users log on, they are taken to the Clinician Prescribing Feedback Report. Figure 2 shows a mockup of the real site with fictitious data. From there, users can drill down into any of several 2012 monthly medication prescribing categories, such as the 'Number of Patients on Non-Formulary Medications in Primary Care', or the 'Number of Patients on a Brand Medication when a Generic is Available'. Users can also drill down into one of several cost categories like 'Opioids' or 'Atypical Antipsychotics'.

Each one of the figures in the monthly columns of the Clinician Prescribing Feedback Report is actually an HTML link to an HBAR or VBAR chart like the one shown in Figure 3. So, depending on which category you'd like to investigate, a user can start at a region-wide view and then drill down all the way to specific provider lists, as seen in Figure 6.

Figure 1

The screenshot shows a web interface titled "Clinician Prescribing Feedback". It features two main sections: a login area and a "Request Access" area. The login area includes input fields for "NUID" and "Password", and a "Login" button. The "Request Access" area includes input fields for "NUID" and "Full Name", and a "Send Access Request" button.

Figure 2

KPNW - 2012 Clinician Prescribing Feedback Report													
Click on monthly values to drill down to Clinic or Department Level													
	Baseline	Jan	Feb	Mar	Apr	May	Jun	Jul	Aug	Sep	Oct	Nov	Dec
<i>Non-Formulary Drugs</i>													
NF Patients for Primary Care	1,262	24	3,733	10,396	2,516	5,508	3,263	8,374	10,233	12,634	7,129	4,852	18,978
NF Patients for Specialty Care	14,409	15,867	24,633	7,716	792	2,160	6,735	22,435	510	1,735	3,908	2,909	14,243
<i>Brands when Generics Available Drugs</i>													
Patients on Brands when Generics Available for Primary Care	7,133	5,190	652	11,174	2,314	42,444	3,849	42,858	8,205	474	21,070	40,867	8,357
Patients on Brands when Generics Available for Specialty Care	3,730	887	6,317	45,987	1,396	11,328	3,974	424	1,438	10,258	4,128	48	2,885
<i>DUAT Priority Drug Class Drugs</i>													
Opioids	\$3,131	\$1,123	\$2,416	\$7,822	\$1,963	\$2,722	\$12,835	\$7,147	\$7,033	\$5,647	\$4,043	\$21,738	\$26,985
Atypical Antipsychotics	\$11,342	\$3,186	\$9,423	\$30,467	\$5,195	\$3,350	\$9,643	\$2,139	\$29,954	\$3,098	\$792	\$8,684	\$31,829
Antidepressants	\$553	\$3,629	\$205	\$36,835	\$6,633	\$10,401	\$9,108	\$9,008	\$2,972	\$9,903	\$16	\$1,525	\$22,322
ADHD	\$12,728	\$11,318	\$18,003	\$3,107	\$4,614	\$4,921	\$6,025	\$7,349	\$13,396	\$13,915	\$5,988	\$2,338	\$2,356
Antiasthmatics & Bronchodilators	\$13,753	\$8,769	\$7,820	\$17,490	\$29,323	\$6,693	\$8,326	\$2,798	\$12,293	\$5,195	\$18,352	\$1,774	\$6,023

For questions about this report contact Michael Nash
 Michael.K.Nash@kp.org 32-7941

PRESCRIBING DATA

Now, that I've shown the concept behind the Clinician Prescribing Feedback site, I will show an example of how my SAS code builds the charts. I will show an example of building charts for Atypical Antipsychotic medications in December 2012. Let's start with the data. As I mentioned earlier, the data in this paper is all fictitious but it mimics the actual data in the site. Since I couldn't use the names of KPNW's clinics I created new Primary Care clinic location names from the names of Oregon counties. Health Record Numbers (HRN) are included in the actual site but have been all changed to 99999999 in this paper. Table 1 describes the data used to create the drill down graphs.

Table 1- SAS Variable Formats in SGF.AA_201212

Variable	Type	Len	Format	Label	Example
HRN	Num	8	9	HRN	99999999
Provider_class	Char	3	\$3.00		PC
PROD_NAME	Char	50	\$50.00	Product Name	ABILIFY TAB 30MG
product	Char	25	\$25.00		ABILIFY
cost	Num	8	DOLLAR10.		\$34
loc_name	Char	25	\$75.00	Clinic	CLATSOP
loc_abbr	Char	12	\$12.00	LOC_ABBR	CLT
dept	Char	25	\$25.00	Department	Family Medicine
dept_abbr	Char	10	\$16.00	Dept.	FM
doc_name	Char	20			BALDWIN_RYAN
prescribing_provider	Char	40	\$40.00	Prescriber	BALDWIN, RYAN
order_type	Char	9	\$15.00	Order Type	Refill
groupdrill	Char	100			href="AA_MH_COST_PC_201212.html"
clinicdrill	Char	100			href="AA_MH_PC_CLT_201212.html"
docdrill	Char	100			href="AA_MH_PC_CLT_BALDWIN_RYAN_201212.html"
yr_mon	Char	6	\$6.00	YR_MON	201212
Lname	Char	14	\$14.00	Last Name	BALDWIN

CODE TO CREATE DRILL DOWN GRAPHS

The code below creates the main SAS data set SGF.AA_201212 that will be used in all the GCHART Procedures to create hbar or vbar graphs for the prescribing of Atypical Antipsychotic products in the region, clinics, and specialty care departments.

```

data SGF.AA_201212;
length provider_class $3. product dept loc_name $25. groupdrill clinicdrill
docdrill $100. prescribing_provider $40. dept_abbr $10.;
set pharm.aa;
FORMAT product DEPT $25. provider_class $3. order_type $15. cost dollar10.;
if provider_class in ('PC') then do;
groupdrill
='href="'||"AA_MH_COST_PC_"||left(trim(put(yr_mon,$10.)))||'.html"';

```

```

clinicdrill='HREF="'||"AA_MH_PC_"||left(trim(put(loc_abbrev,$10.))||'_'||left(
trim(put(yr_mon,$10.))||'.html"');
docdrill
='HREF="'||"AA_MH_PC_"||left(trim(put(loc_abbrev,$20.))||'_'||left(trim(put(do
c_name,$50.))||'_'||left(trim(put(yr_mon,$10.))||'.html"');
end;
else if provider_class = 'SC' then do;
groupdrill
='HREF="'||"AA_MH_COST_SC_"||left(trim(put(yr_mon,$10.))||'.html"');

clinicdrill='HREF="'||"AA_MH_SC_"||left(trim(put(dept_abbrev,$10.))||'_'||left(
(trim(put(yr_mon,$10.))||'.html"');
docdrill
='HREF="'||"AA_MH_SC_"||left(trim(put(dept_abbrev,$20.))||'_'||left(trim(put(d
oc_name,$50.))||'_'||left(trim(put(yr_mon,$10.))||'.html"');
end;
attrib prescribing_provider label='Prescriber'
prod_name label = 'Product Name'
order_type label = 'Order Type'
loc_name label = 'Clinic'
dept label = 'Department'
dept_abbrev label = 'Dept.';
run;

proc sort data = SGF.AA_201212;
by yr_mon;
run;

goptions reset=all ROTATE=landscape device=gif border;
options symbolgen macrogen;

PATTERN1 VALUE = SOLID COLOR = BLUE;
PATTERN2 VALUE = SOLID COLOR = ORANGE;
PATTERN3 VALUE = SOLID COLOR = PURPLE;
PATTERN4 VALUE = SOLID COLOR = BROWN;
PATTERN5 VALUE = SOLID COLOR = CREAM;
PATTERN6 VALUE = SOLID COLOR = DarkTurquoise;
PATTERN7 VALUE = SOLID COLOR = GOLD;
PATTERN8 VALUE = SOLID COLOR = GRAY;
PATTERN9 VALUE = SOLID COLOR = STEEL;
PATTERN10 VALUE = SOLID COLOR = LILAC;

```

My code works by letting the data in SGF.AA_201212 build the graphs for me. My code includes four macro call routines (%REGION, %GROUP, %CLINIC, and %DOC). The %REGION macro call routine builds the graph in Figure 3, prescribing for the entire region. In the PROC GCHART procedure within the %REGION macro is the HTML= option which is set to the variable 'GROUPDRILL', which is the html page for prescribing for all the clinics (Figure 4) that is built in the next macro call routine named %GROUP. The HTML= option in the %GROUP macro is set to 'CLINICDRILL', which is the html page for prescribing of each individual Primary Care clinic, or Specialty Care Department like the one for the Clatsop Primary Care Clinic in Figure 5.

So you can see that each bar in each graph is tied to the corresponding data in the 'SGF.AA_201212' data set and is linked to an html page by the HTML= option via the variables: groupdrill, clinicdrill, and docdrill. Below, Table 2 describes the functions of each macro call routine and shows what HTML= variable is used to drill down to the next level.

Table 2

Macro call routine	Does what?	HTML= variable	Bars drill down to:	Example
%REGION	Compares prescribing within the region by provider class (Primary Care or Specialty Care)	GROUPDRILL	All Primary Care Clinics or All Specialty Care depts	Figure 3
%GROUP	Compares prescribing within Primary Care by clinic or within Specialty Care by dept.	CLINICDRILL	Each Primary Care Clinic or each Specialty Care dept	Figure 4
%CLINIC	Compares prescribing among doctors within each Primary Care clinic or Specialty Care dept.	DOCDRILL	Each doctor	Figure 5
%DOC	Builds member-level lists for each doctor.	N/A	N/A	Figure 6

Create Region Level Charts (%REGION)

The code below creates the chart in Figure 3.

```

%macro region(region1);

ods listing close;

* Bar chart of Atypical Antipsychotic costs by provider class ;
* (Primary Care or Specialty Care) ;

legend1 label=none frame;

FILENAME ODSOUT "C:\SGF 2013\&region1.";
ods html body="AA_MH_COST_&region1..html" path=odsout ;

AXIS1 LABEL = NONE value = ('Primary Care' 'Specialty Care');
axis2 label=(angle=0 "Monthly Cost") minor=(n=1) offset=(0,0);
axis3 label=none ;

title1 j=c "Cost for Atypical Antipsychotics for &region1.";
title2 j=C 'Click on bars to drill down further';

proc gchart data=SGF.AA_201212 GOUT = GRAPH;
  hbar provider_class / discrete
    sumvar=cost
    html=groupdrill
    SUBGROUP=product
    sum
    type=sum
    cframe=grayaa

```

```

space=0
width=6
gspace=5
maxis=axis1
raxis=axis2
gaxis=axis3
name="AA_MH_COST_&region1."
outline=black
legend=legend1;
where yr_mon="&region1.";
format cost dollar14.;

run;
quit;


ods html close;

%mend region;

data _null_;
  set SGF.AA_201212;
  by yr_mon;
  if first.yr_mon then
call execute('%region(' || trim(yr_mon) || ');');
run;

```

Figure 3 (Prescribing for the entire KPNW Region)



Create Charts for all Clinics or Specialty Care Departments (%GROUP)

The code below creates the chart in Figure 4.

```

%macro group(grp1,grp2,grp3,grp4);
* Compare all Primary Care Clinics in one graph. ;
* Do same for all Specialty Care Depts *;
* grp1 = PC or SC;
* grp2 = loc_name (PC) or dept_abbr (SC);
* grp3 = yr_mon ;
* grp4 = Primary Care or Specialty Care ;

ods html body="AA_MH_cost_&grp1._&grp3..html" path=odsout;

AXIS1 LABEL = NONE VALUE = (ANGLE = -65 ROTATE = 0);
axis2 label=(angle=0 "Monthly Cost") minor=(n=1) offset=(0,0);
axis3 label=none;
AXIS4 LABEL = NONE VALUE = (ANGLE = 0 ROTATE = 0);

title1 j=c "Costs for Atypical Antipsychotics by &grp4. for &grp3.";
title2 j=C 'Click on bars to drill down further';

proc gchart data=SGF.AA_201212 GOUT = GRAPH;
  hbar &grp2. / discrete
              sumvar=cost
              html=clinicdrill
              SUBGROUP=product
              sum
              type=sum
              cframe=grayaa
              space=0
              width=6
              gspace=5
              maxis=axis4
              raxis=axis2
              gaxis=axis3
              name="AA_MH_cost_&grp1._&grp3."
              coutline=black
              legend=legend1;
where provider_class="&grp1." and yr_mon = "&grp3.";
format cost dollar14.;
run;quit;

ods html close;


%mend group;

data _null_;
  set SGF.AA_201212 (where=(provider_class='PC'));
  if _n_ = 1;
call execute('%group(' ||trim(provider_class) ||','|| 'loc_name' ||','||
|| trim(yr_mon) || ','|| 'Primary Care Clinics' || ');');
run;

data _null_;
  set SGF.AA_201212 (where=(provider_class='SC'));
  if _n_ = 1;
call execute('%group(' ||trim(provider_class) ||','|| 'dept_abbr' ||','||
|| trim(yr_mon) || ','|| 'Specialty Care Depts.' || ');');
run;

```

Figure 4 (Prescribing of all Clinics in Primary Care)



Create Charts for each Clinic or Specialty Care Department (%CLINIC)

The code below creates the chart in Figure 5.

```
%macro clinic(clinic1,clinic2,clinic3,clinic4,clinic5,clinic6,clinic7);

* Create graphs for each Primary Care Clinic or each Specialty Care Dept *;
* clinic1 = PC or SC;
* clinic2 = loc_abbr (PC) or dept_abbr (SC);
* clinic3 = Clinic or Dept abbr.;
* clinic4 = Clinic or Dept Name;
* clinic5 = hbar or vbar ;
* clinic6 = axis1 (for vbar) or axis4 (for hbar) ;
* clinic7 = yr_mon ;

ods html body="AA_MH_&clinic1._&clinic3._&clinic7..html" path=odsout;

AXIS1 LABEL = NONE VALUE = (ANGLE = -65 ROTATE = 0);
axis2 label=(angle=0 "Monthly Cost") minor=(n=1) offset=(0,0);
axis3 label=none;
AXIS4 LABEL = NONE VALUE = (ANGLE = 0 ROTATE = 0);
```

```

title1 j=c "Costs for Atypical Antipsychotics for &clinic1. providers at
&clinic4. for &clinic7.";
title2 j=C 'Click on bars to drill down further';

proc gchart data=SGF.AA_201212 GOUT = GRAPH;
  &clinic5. lname / discrete
    sumvar=cost
    html=docdrill
    SUBGROUP=product
    sum
    type=sum
    cframe=grayaa
    space=0
    width=6
    gspace=5
    maxis=&clinic6.
    raxis=axis2
    gaxis=axis3
    name="AA_MH_&clinic1._&clinic3._&clinic7."
    outline=black
    legend=legend1;
where provider_class="&clinic1." and &clinic2. = "&clinic3.";
format cost dollar14.;
run;
quit;

ods html close;

%mend clinic;

proc sort data = SGF.AA_201212 (where=(provider_class='PC')) nodupkey
out=pc_nodup;
  by loc_abbr;
run;


proc sort data = SGF.AA_201212 (where=(provider_class='SC')) nodupkey
out=sc_nodup;
  by dept_abbr;
run;

data _null_;
  set work.PC_nodup;
  by provider_class loc_abbr;
  if first.loc_abbr then
call execute('%clinic(' || trim(provider_class) || ', ' || 'loc_abbr' || ', ' ||
trim(loc_abbr) || ', ' || trim(loc_name) || ', ' || 'hbar' || ', ' || 'axis4' || ', ' ||
trim(yr_mon) || ')');
run;

data _null_;
  set work.SC_nodup;
  by provider_class dept_abbr;
  if first.dept_abbr then
call execute('%clinic(' || trim(provider_class) || ', ' || 'dept_abbr' || ', ' ||
trim(dept_abbr) || ', ' || trim(dept) || ', ' || 'hbar' || ', ' || 'axis4' || ', ' ||
trim(yr_mon) || ')');
run;

```

Figure 5 (Prescribing by doctors within a Clinic)



Create Patient Lists for each Doctor (%DOC)

The code below creates the patient lists in Figure 6.

```
%macro doc(doc1,doc2,doc3,doc4,doc5,doc6,doc7,doc8);

* Create lists for each doctor *;
* doc1 = PC, SC *;
* doc2 = loc_abbr (for PC), dept_abbr (for SC) *;
* doc3 = loc_name (for PC) , dept (for SC) *;
* doc4 = hbar, vbar *;
* doc5 = axis1 or axis4 *;
* doc6 = lname *;
* doc7 = doc_name ;
* doc8 = yr_mon ;

ods html body="AA_MH_&doc1._&doc2._&doc7._&doc8..html" path=odsout;

proc print data=SGF.AA_201212 label noobs;
  where provider_class="&doc1." and doc_name ="&doc7." ;
  var hrn loc_name prescribing_provider prod_name order_type cost;
  label loc_name = 'Clinic'
```

```

prescribing_provider = 'Prescriber'
order_type = 'Order Type'
prod_name = 'Product Name';
title1 "Costs for Atypical Antipsychotics dispensed by &doc6. in &doc8.";
format cost dollar10.;
run;

ods html close;
ods listing;

%mend doc;

proc sort data = SGF.AA_201212 (where=(provider_class='PC'))
  nodupkey out=pc_doc_nodup;
  by provider_class loc_abbr prescribing_provider;
run;

data _null_;
  set work.pc_doc_nodup;
  by provider_class loc_abbr prescribing_provider;
  if first.prescribing_provider then
call execute('%doc(' || trim(provider_class) || ', ' ||
trim(loc_abbr) || ', ' || trim(loc_name) || ', ' || 'hbar' || ', ' || 'axis4' || ', ' ||
trim(lname) || ', ' || trim(doc_name) || ', ' || trim(yr_mon) || ')');
run;

proc sort data = SGF.AA_201212 (where=(provider_class='SC')) nodupkey
out=sc_doc_nodup;
  by provider_class loc_abbr prescribing_provider;
run;

data _null_;
  set work.sc_doc_nodup;
  by provider_class loc_abbr prescribing_provider;
  if first.prescribing_provider then
call execute('%doc(' || trim(provider_class) || ', ' ||
trim(dept_abbr) || ', ' || trim(dept) || ', ' || 'hbar' || ', ' || 'axis4' || ', ' ||
trim(lname) || ', ' || trim(doc_name) || ', ' || trim(yr_mon) || ')');
run;

```

Figure 6 (List of patients for Dr. Baldwin)

Costs for Atypical Antipsychotics dispensed by BALDWIN in 201212

HRN	Clinic	Prescriber	Product Name	Order Type	cost
99999999	CLATSOP	BALDWIN, RYAN	ABILIFY TAB 30MG	Refill	\$34

CONCLUSION

The Clinician Prescribing Feedback Site is a secure intranet site where pharmacists, doctors, and pharmacy managers can track prescribing of certain drug categories or therapeutic classes. My code uses PROC GCHART within an ODS HTML statement to create HBAR or VBAR charts to compare medication prescribing across an organization, a clinic, or a specialty. I link the charts via the HTML= option so users can drill down to a specific clinic or provider. And I provide patient lists to the users so the information is timely and actionable. My code is specifically tailored for pharmaceutical data but the overall concept is fairly simple and can be easily applied to other industries.

ACKNOWLEDGMENTS

The author would like to thank Steve Piscoran for his help with data mining and Prasad S. Ravi for his SAS coding expertise.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Michael K. Nash
Kaiser Permanente Northwest
5725 NE 138th Avenue
Portland, OR 97230-3409
Phone: (503) 261-7941
Fax: (503) 261-7567
E-mail: Michael.K.Nash@kp.org

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.